

Lesson 9

Integration by Parts

Initializations

```
> restart;  
with(student):  
>
```

9.1 Integration by Parts for Definite and Indefinite Integrals

Integration by parts is described by the formula

$$\int u \, dv = uv - \int v \, du$$

The power of this method lies in the fact that it effectively differentiates u , which has the potential of simplifying the integrand. Mathematical details will be provided in class.

Maple syntax for integration by parts is:

intparts(integral expression, part of the integrand to be differentiated);

The **intparts** routine is located in the **student** package.

Examples

Example 9.1.1.

Compute $\int x \sin x \, dx$.

Solution

Differentiation can remove the polynomial term x , so we choose $u = x$.

```
> e1:=Int(x*sin(3*x), x);  
e2:=intparts(e1, x);
```

$$e1 := \int x \sin(3x) \, dx$$

$$e2 := -\frac{1}{3} x \cos(3x) - \left(\int \left(-\frac{1}{3} \cos(3x) \right) dx \right) \quad (2.1.1.1)$$

Let Maple evaluate the remaining elementary integral.

```
> answer:=value(e2)+C;
```

$$answer := -\frac{1}{3} x \cos(3x) + \frac{1}{9} \sin(3x) + C \quad (2.1.1.2)$$

```
>
```

Example 9.1.2

Evaluate $\int_0^3 x^2 e^{-4x} dx$.

Solution

Use integration by parts twice to eliminate the x^2 term.

```
> e1:=Int(x^2*exp(-4*x), x=0..3);  
e2:=intparts(e1, x^2);
```

$$e1 := \int_0^3 x^2 e^{-4x} dx$$

$$e2 := -\frac{9}{4} e^{-12} - \left(\int_0^3 \left(-\frac{1}{2} x e^{-4x} \right) dx \right) \quad (2.1.2.1)$$

```
> e3:=intparts(e2, x);
```

$$e3 := -\frac{21}{8} e^{-12} + \int_0^3 \frac{1}{8} e^{-4x} dx \quad (2.1.2.2)$$

Let Maple evaluate the remaining elementary integral.

```
> answer:=value(e3);  
evalf(answer);
```

$$answer := -\frac{85}{32} e^{-12} + \frac{1}{32} \\ 0.03123367944 \quad (2.1.2.3)$$

```
>
```

▼ **Example 9.1.3**

Compute $\int e^{2x} \cos 3x dx$.

Solution

In this example we have to integrate by parts twice and both times we will differentiate the exponential function.

```
> e1:=Int(exp(2*x)*cos(3*x), x);  
e2:=intparts(e1, exp(2*x));
```

$$e1 := \int e^{2x} \cos(3x) dx$$

$$e2 := \frac{1}{3} e^{2x} \sin(3x) - \left(\int \frac{2}{3} e^{2x} \sin(3x) dx \right) \quad (2.1.3.1)$$

```
> e3:=intparts(e2, exp(2*x));
```

$$e3 := \frac{1}{3} e^{2x} \sin(3x) + \frac{2}{9} e^{2x} \cos(3x) + \int \left(-\frac{4}{9} e^{2x} \cos(3x) \right) dx \quad (2.1.3.2)$$

Observe that the original integral came back and we are now faced with the equation:

```
> e4:=simplify(e1=e3);
```

$$e4 := \int e^{2x} \cos(3x) dx = \frac{1}{3} e^{2x} \sin(3x) + \frac{2}{9} e^{2x} \cos(3x) - \frac{4}{9} \int e^{2x} \cos(3x) dx \quad (2.1.3.3)$$

We solve this equation for $\int e^{2x} \cos 3x dx$. Since the indefinite integrals on the left and right hand side may differ a constant, that constant needs to be introduced at this time.

> e5:=isolate(e4, e1)+(0=C);

$$e5 := \int e^{2x} \cos(3x) dx = \frac{3}{13} e^{2x} \sin(3x) + \frac{2}{13} e^{2x} \cos(3x) + C \quad (2.1.3.4)$$

>